Límites y continuidad de funciones

Los objetivos de esta práctica son:

- Calcular límites de funciones reales de una variable real.
- Dibujar gráficas sencillas de funciones.
- Estudiar la continuidad de funciones reales de una variable real.
- Comprender gráficamente el significado de la continuidad de una función en un punto (ε, δ).
- Localizar ceros de funciones aplicando el teorema de Bolzano.
- Hacer sencillos ejercicios de programación con los conceptos anteriores.

1. Límites de funciones. Continuidad

1) Calcular $\lim_{x \to 1} \left(\frac{1}{1-x} - \frac{3}{1-x^3} \right)$ y representar gráficamente la función y el límite.

En Maple, la orden para calcular límites de funciones es muy sencilla. Obsérvese que para indicar "$x \to 1$" , lo que realmente se escribe es **x=1**. Esto es solo una forma de escribir en Maple.

```
> limit(1/(1-x)-3/(1-x^3),x=1);
```

$$-1$$

La orden **Limit** representa el mismo límite, pero sin calcular su valor.
La orden **limit(...)** equivale a **value(Limit(...))**:

```
> value(Limit(1/(1-x)-3/(1-x^3),x=1));
```

$$-1$$

Podemos aprovechar las dos órdenes para que quede escrito más claro:

```
> Limit(1/(1-x)-3/(1-x^3),x=1)=limit(1/(1-x)-3/(1-x^3),x=1);
```

$$\lim_{x \to 1} \left(\frac{1}{1-x} - \frac{3}{1-x^3} \right) = -1$$

Ahora representemos la función. Dibujamos la gráfica para x en el intervalo $[0, 2]$ e y en $[-2, 0]$, por ejemplo. Añadimos la recta horizontal $y = -1$, que nos ayuda a ver mejor el límite. Pedimos que la función se dibuje en rojo y la recta $y = -1$ en verde. Y en lugar de las etiquetas "x" e "y" que aparecerían junto a los ejes, podemos escribir cualquier otra cosa, o incluso nada:

```
> plot([1/(1-x)-3/(1-x^3),-1],x=0..2,y=-2..0,color=[red,blue],
  labels=["",""]);
```



2) Calcular $\lim_{x \to \infty} (\log(x) - \log(2x-1))$ y representar gráficamente la función y el límite.

Podemos calcular también límites en ∞ (y en −∞):

```
> limit(log(x)-log(2*x-1),x=infinity);
```

$$-\ln(2)$$

Y representar gráficas de funciones en intervalos no acotados. Por ejemplo, la función anterior en el intervalo [1, ∞). Hay que acostumbrarse a interpretar la respuesta adecuadamente.

```
> plot([log(x)-log(2*x-1),-log(2)],x=1..infinity,y=-1..0,
  color=[red,blue]);
```



Observemos cómo Maple no representa bien la función $\log(x) - \log(2x-1)$ para valores grandes de x (es decir, cuando $x \to \infty$). Esto se debe probablemente a errores de redondeo, ya que se manejan, de modo aproximado, cantidades muy grandes que luego se restan.

A continuación representamos la misma función, expresada de otra forma. Ahora la representación gráfica es más fiel.

```
> plot([log(x)-log(x-1/2)-log(2),-log(2)],x=1..infinity,y=-1.
  .0,color=[red,blue]);
```



3) Sea $f(x) = 0$ si $x = 0$ y $f(x) = \dfrac{1}{1 + 2^{\frac{1}{x}}}$ en otro caso.

Estudiar la continuidad de esta función en $x = 0$ y representarla gráficamente en un entorno de 0.

```
> restart;
  f:=x->piecewise(x=0,0,1/(1+2^(1/x))): f(x);
  limit(f(x),x=0);
```

$$\begin{cases} 0 & x = 0 \\ \dfrac{1}{1 + 2^{\frac{1}{x}}} & \text{otherwise} \end{cases}$$

$$\textit{undefined}$$

Según Maple, no existe el límite, así que la función no es continua en 0. Podemos averiguar algo más; por ejemplo, podemos calcular los límites laterales con las opciones **left** y **right**:

```
> [Limit('f(x)',x=0,left)=limit(f(x),x=0,left),
  Limit('f(x)',x=0,right)=limit(f(x),x=0,right)];
```

$$\left[\lim_{x \to 0^-} f(x) = 1,\ \lim_{x \to 0^+} f(x) = 0\right]$$

Así que los límites laterales son distintos y, en efecto, no hay límite en 0. La función sí es continua por la derecha, pero no por la izquierda.

Dibujemos la gráfica de la función en un entorno de 0. La orden **tickmarks=[m,n]** indica cuántas divisiones queremos hacer en cada eje; pero Maple no siempre lo tiene en cuenta.

```
> plot(f(x),x=-1/2..1/2,tickmarks=[10,5],thickness=2);
```



Se observa un trazo rojo vertical en el eje y. Esto se debe a que, para dibujar una gráfica, Maple dibuja primero unos cuantos puntos aislados y luego los une con trazo continuo. Así, une los puntos (0,1) y (0,0), aproximadamente. Podemos evitarlo añadiendo la opción **discont=true**, que indica que la función puede no ser continua en algún punto:

```
> plot(f(x),x=-1/2..1/2,tickmarks=[10,5],thickness=2,
  discont=true);
```



2. La definición de continuidad

4) Sea la función $f(x) = \dfrac{x+2}{x-2}$. Para $b = 3$ y $\varepsilon = \dfrac{1}{10}$, calcular gráficamente δ.

```
> restart;f:=x->(x+2)/(x-2):b:=3:epsilon:=1/10:
```

Dibujamos la función y el valor $f(b)$, para x en un intervalo $(b - \delta, b + \delta)$ e y entre $f(b) - \varepsilon$ y $f(b) + \varepsilon$. Probamos con diferentes valores de δ hasta dar con uno apropiado (obsérvese la opción **axes=boxed**), de modo que la gráfica no se salga del recuadro por arriba ni por debajo:

```
> delta:=0.02;
  plot([f(x),f(b)],x=b-delta..b+delta,
  y=f(b)-epsilon..f(b)+epsilon,axes=boxed,color=[red,blue]);
```

$$\delta := 0.02$$



El valor $\delta = 0.02$ es suficiente.

3. Teorema de Bolzano

5) La función $f(x) = x + \frac{1}{x}$ cumple que $f(1) = 2$ y $f(-1) = -2$. ¿Existe algún punto x en $[-1, 1]$ en el que $f(x) = 0$? ¿Contradice esto el teorema de Bolzano?

La función no es continua en $[-1, 1]$, en concreto, no lo es en 0. Podemos verlo en la gráfica:

```
> plot(x+1/x,x=-1..1,y=-10..10,discont=true);
```



Y como no es continua en todo el intervalo, no se le puede aplicar el teorema de Bolzano. También se observa en la gráfica que no existe ningún punto x en $[-1, 1]$ en el que $f(x) = 0$. Esto también se puede comprobar resolviendo la ecuación x + 1/x = 0 :

```
> solve({x+1/x=0});
```

$$\{x = \mathrm{I}\}, \{x = -\mathrm{I}\}$$

Han salido dos soluciones, pero no son números reales, sino los números complejos $-i, i$. Si solo queremos las soluciones reales (números complejos x con parte imaginaria nula), Maple no devuelve ninguna:

```
> solve({x+1/x=0,Im(x)=0});
Warning, solutions may have been lost
```

6) Probar que la ecuación $f(x) = x^{179} + \frac{163}{1 + x^2 + sen(x)^2} = 119$ tiene al menos 1 solución real.

f es continua en **R** (el denominador nunca se anula). Además, $f(0) = 163$ y $f(1) < 1 + \frac{163}{2}$. Por el teorema de Bolzano, en el intervalo (0, 1) existe algún x tal que $f(x) = 119$. Es fácil probar, de manera análoga, que hay alguna otra solución en el intervalo $(-1, 0)$.

```
> plot([x^179+163/(1+x^2+sin(x)^2),119],x=-1..1,
  color=[red,blue]);
```



7) Demostrar que la ecuación $sen(x) = x - 1$ tiene al menos una solución real.

Si definimos la función $f(x) = sen(x) - x + 1$, se trata de probar que existe algún número real x tal que $f(x) = 0$. Pero la función es continua, $f(0) = 1$ y $f(3) = sen(3) - 2 < 0$. Por el teorema de Bolzano, existe al menos una solución real de la ecuación $f(x) = 0$.

```
> plot([sin(x),x-1,sin(x)-(x-1)],x=-infinity..infinity,
  color=[red,blue,green],thickness=[1,1,2]);
```



4. Programación

Búsqueda de soluciones por intervalos encajados

El siguiente procedimiento busca una solución de la ecuación $f(x)=0$ en el intervalo $[a, b]$, con una precisión ε y presentando n cifras significativas. La sintaxis es **bolzano(f,a,b, epsilon,n)**, donde f tiene que ser de tipo "procedure", a, b, ε son números reales y n es un número natural.

El método es el de los intervalos encajados: se supone que $f(a)$ y $f(b)$ tienen distinto signo (de lo contrario, se da un mensaje de error) y se hacen sucesivas divisiones del intervalo $[a, b]$ de modo que los extremos tengan siempre distinto signo y la longitud sea la mitad que en el paso anterior. Cuando la longitud del intervalo es menor que ε, se proporciona como solución aproximada el punto medio del intervalo, con n cifras significativas.

Si a o b son soluciones exactas, el procedimiento lo dice.

Naturalmente, para que el método sea fiable la función tiene que ser continua.

Dado que f tiene que ser de tipo "procedure", no es correcto escribir **bolzano(x^6+x-1, 0,4,10^(-20),20)**. Si queremos encontrar una solución aproximada de la ecuación $x^6+x-1=0$ en el intervalo $[0, 4]$, debemos definir antes $f(x)=x^6+x-1$ y luego escribir **bolzano(f,0,4,10^(-20),20)**, por ejemplo.

```
> bolzano:=proc(f::procedure,a0,b0,eps,n)
  local a,b,medio;
  if f(a0)=0 and f(b0)=0 then {a0,b0};
     elif f(a0)=0 then a0;
     elif f(b0)=0 then b0;
     elif sign(evalf(f(a0),n))=sign(evalf(f(b0),n)) then
          error "no se puede aplicar el teorema de Bolzano,
  porque sign(%1(%2))=sign(%1(%3))",f,a0,b0;
  else
    a:=a0:
    b:=b0:
    while evalf(b-a)>eps do
      medio:=(a+b)/2;
      if sign(evalf(f(a),n))=sign(evalf(f(medio),n)) then
  a:=medio;
         else b:=medio;
      fi;
    od;
    evalf((a+b)/2,n);
  fi;
  end:
```

8) Calcular una solución aproximada de la ecuación $x^6+x-1=0$ en el intervalo $[0, 4]$, con una precisión de 10^{-20} y presentarla con 20 cifras significativas.

```
> unassign('f'):
  f:=x->x^6+x-1:
  bolzano(f,0,4,10^(-20),20);
```

$$0.77808959867860109788$$

9) Calcular una solución aproximada de la ecuación $sen(x)=0$ en el intervalo $[3, 5]$, con una precisión de 10^{-31} y presentarla con 30 cifras significativas. Compararla con el número π.

```
> bolzano(sin,3,5,10^(-31),30);
```

$$3.14159265358979323846264338328$$

```
> evalf(Pi,30);
```

$$3.14159265358979323846264338328$$